Wprowadzenie do Gnuplota

Janusz Szwabiński

Gnuplot

- strona główna: `http://gnuplot.info/`
- program do tworzenia wykresów 2D i 3D
- darmowy
- rozwijany od 1986 roku
- działa w trybie poleceń
- dwa tryby pracy: interaktywny i wsadowy
- wykorzystywany jako silnik graficzny przez liczne programy (np. GNU Octave, Maxima)
- dostępny na wielu systemach operacyjnych
- w połączeniu z programami powłoki uniksowej (np. ‘sed‘, ‘awk‘) pozwala na automatyzację zadań związanych z tworzeniem wielu podobnych wykresów
- czasami wykorzystywany jako prosty silnik do wizualizacji w czasie rzeczywistym

Pierwsze kroki

```
szwabin@voyager:~/Dropbox/Zajęcia/Matematyka/Gnuplot$ gnuplot

G N U P L O T
Version 4.6 patchlevel 4 last modified 2013-10-02
Build System: Linux x86_64

Copyright (C) 1986-1993, 1998, 2004, 2007-2013
Thomas Williams, Colin Kelley and many others

gnuplot home:   http://www.gnuplot.info
faq, bugs, etc: type "help FAQ"
immediate help: type "help" (plot window: hit 'h')

Terminal type set to 'wxt'
gnuplot>
```

Pierwsze kroki

```
gnuplot> help
 ‘Gnuplot‘ is a portable command-line driven graphing utility for Linux, OS/2,
 MS Windows, OSX, VMS, and many other platforms. The source code is copyrighted
 but freely distributed (i.e., you don’t have to pay for it). It was originally
 created to allow scientists and students to visualize mathematical functions
 and data interactively, but has grown to support many non-interactive uses
 such as web scripting. It is also used as a plotting engine by third-party
 applications like Octave. Gnuplot has been supported and under active
 development since 1986.

 Gnuplot supports many types of plots in either 2D and 3D. It can draw using
 lines, points, boxes, contours, vector fields, surfaces, and various
 associated text. It also supports various specialized plot types.

 Gnuplot supports many different types of output: interactive screen terminals
 (with mouse and hotkey input), direct output to pen plotters or modern
 printers, and output to many file formats (eps, emf, fig, jpeg, LaTeX, pdf, png,
 postscript, ...). Gnuplot is easily extensible to include new output modes.
 Recent additions include interactive terminals based on wxWidgets (usable
 on multiple platforms), and Qt. Mouseable plots embedded in web pages
 can be generated using the svg or HTML5 canvas terminal drivers.

 The command language of ‘gnuplot‘ is case sensitive, i.e. commands and
Press return for more:
```

Czy jest jakieś GUI?





Pierwszy wykres

```
gnuplot> plot sin(x)
```



Coś bardziej skomplikowanego

```
gnuplot> a = 0.9 # amplituda
gnuplot> f(x) = a*sin(x) # funkcja użytkownika
gnuplot> g(x) = a*cos(x)
gnuplot> plot f(x) title 'sin(x)' with lines,\
> g(x) title 'cos(x)' with points
```



Modyfikacja wyglądu

```
gnuplot> set border linewidth 1.5 # grubość osi
gnuplot> set style line 1 linecolor rgb '#0060ad' linetype 1 linewidth 2
gnuplot> set style line 2 linecolor rgb '#dd181f' linetype 1 linewidth 2
gnuplot> set xrange [-10:10]      # zakres na osi X
gnuplot> set yrange [-1.1:1.1]
gnuplot> set title 'Funkcje trygonometryczne' # tytuł wykresu
gnuplot> set xlabel 'x' # etykieta osi
gnuplot> set ylabel 'y'
gnuplot> plot f(x) title 'sin(x)' with lines linestyle 1,\
> g(x) title 'cos(x)' with lines linestyle 2
```



Modyfikacji ciąg dalszy

```
gnuplot> set key at 6.1,1.3 # wstaw legendę w konkretnym punkcie
gnuplot> set xrange [-2*pi:2*pi]
gnuplot> set yrange [-1.5:1.5]
gnuplot> set xtics ('-2pi' -2*pi, '-pi' -pi, 0, 'pi' pi, '2pi' 2*pi)
gnuplot> set ytics 1 # co jeden
gnuplot> set tics scale 0.75
gnuplot> plot f(x) title 'sin(x)' with lines ls 1, \
>     g(x) title 'cos(x)' with lines ls 2
```



Wykresy funkcji dwóch zmiennych

```
gnuplot> set pm3d #pokoloruj wykres
gnuplot> splot [x=-3:3] [y=-3:3] sin(x)*cos(y)
```



Prezentacja danych na wykresie

```
# plik dane1.dat
# X   Y
  1   2
  2   3
  3   2
  4   1
```

Prezentacja danych na wykresie

```
gnuplot> plot 'dane1.dat'
```



Prezentacja danych na wykresie

```
gnuplot> set style line 1 lc '#0060ad' lt 1 lw 2 pt 7 ps 1.5
gnuplot> set xlabel 'x'
gnuplot> set ylabel 'y'
gnuplot> set title 'Wykres z danymi'
gnuplot> set xr [0:5]
gnuplot> set yr [0:4]
gnuplot> plot 'dane1.dat' title 'Eksperyment' with linespoints ls 1
```



Dane obarczone błędem

```
#
# $Id: battery.dat,v 1.1.1.1 1998/04/15 19:16:41 lhecking Exp $
#
50.000000 0.036990 2.500000 0.007039
47.000000 0.036990 2.500000 0.007039
44.000000 0.038360 2.500000 0.007053
41.000000 0.042160 2.500000 0.007050
38.000000 0.043200 2.500000 0.007018
35.000000 0.046900 2.500000 0.007021
32.000000 0.048840 2.500000 0.006963
29.000000 0.052000 2.500000 0.006929
26.000000 0.055470 2.500000 0.006947
23.000000 0.060000 2.500000 0.006882
20.000000 0.064660 2.500000 0.006879
17.000000 0.069600 2.500000 0.006936
14.000000 0.079800 2.500000 0.007080
11.000000 0.086920 2.500000 0.007232
8.000000 0.085500 2.500000 0.007262
5.000000 0.101260 2.500000 0.008415
2.000000 0.091000 2.500000 0.011203
0.000000 0.081480 2.500000 0.011828
```

Dane obarczone błędem

```
gnuplot> set xr [-2:52]
gnuplot> set yr [0:0.12]
gnuplot> set ylabel 'Moc [mW]'
gnuplot> set xlabel 'Opor [Ohm]'
gnuplot> set format y '%.0s'
gnuplot> plot 'battery.dat' using 1:2:4 w yerrorbars ls 1, \
>     '' using 1:2 w lines ls 1
```



Działania na danych

```
gnuplot> set format y
gnuplot> set yrange [0:120]
gnuplot> plot 'battery.dat' using 1:($2*1000):($4*1000) w yerrorbars ls 1
```



Zapisywanie wykresów do plików

- w formacie png

```
gnuplot> set terminal pngcairo size 350,262 enhanced font 'Verdana,10'
Terminal type set to 'pngcairo'
Options are 'background "#ffffff" enhanced font "Verdana,10" fontscale 1.0 size 350, 262'
gnuplot> set output 'test.png'
gnuplot> plot sin(x)
```

- w formacie svg

```
gnuplot> set terminal svg size 350,262 fname 'Verdana' fsize 10
Terminal type set to 'svg'
Options are 'size 350,262 fixed fname 'Verdana' fsize 10 butt solid'
gnuplot> set output 'test.svg'
gnuplot> plot sin(x)
```

- w formacie postscript

```
gnuplot> set terminal postscript eps enhanced color font 'Helvetica,20'
Terminal type set to 'postscript'
Options are 'eps enhanced defaultplex \
   leveldefault color colortext \
   dashed dashlength 1.0 linewidth 1.0 butt noclip \
   nobackground \
   palfuncparam 2000,0.003 \
   "Helvetica" 20 fontscale 1.0 '
gnuplot> set output 'test.eps'
```

Bibliografia

- oficjalna dokumentacja Gnuplota:
  `http://gnuplot.info/documentation.html`
- *Gnuplotting - Create scientific plots using gnuplot*:
  `http://www.gnuplotting.org/`